
Prüfziffer für EAN Nummer berechnen 

1 Aufgabe 

Wie wird die Prüfziffer einer vorgegebenen 12-stelligen EAN1-Nummer (z. B. 401234598765) mit 
VBA-Prozeduren automatisch berechnet? Diese Frage wird im Folgenden beantwortet. Vorher wird 
ein Zahlenbeispiel vorgeführt. 

2 Zahlenbeispiel 

Die Prüfziffer einer EAN-Nummer wird nach dem Modulus-10-Verfahren ermittelt. Die Ziffern der 
EAN-Nummer ohne Prüfziffer werden von rechts nach links mit den Gewichtungsfaktoren 3 und 1 
multipliziert und die Produkte addiert. Die Produktsumme wird ganzzahlig durch 10 geteilt und der 
Divisionsrest wiederum von 10 abgezogen. Das Ergebnis ist die Prüfziffer. Ist das Ergebnis 10, so ist 
die Prüfziffer eine 0 (s. Tabelle 1). 

Datenstellen der EAN z1 z2 z3 z4 z5 z6 z7 z8 z9 z10 z11 z12 z13 
EAN-Nummer 4 0 1 2 3 4 5 9 8 7 6 5 2 
Gewichtung 1 3 1 3 1 3 1 3 1 3 1 3  
Produkte 4 0 1 6 3 12 5 27 8 21 6 15 =108 
Prüfziffer = 10 – 108 mod10 = 2 

Tabelle 1: Prüfziffer für EAN-Nummer berechnen 

Die Produktsumme (108) wird ganzzahlig durch 10 geteilt und der Divisionsrest wiederum von 10 
abgezogen. Das Ergebnis ist die Prüfziffer. 

3 Anwendung 

Die Anwendung besteht aus zwei VBA-Prozeduren: 

Prozedur 

Typ Name Aufgabe 

Subprozedur MainProc Treiber für Prüfziffernberechnung 

Funktionsprozedur GetCheckDigit Prüfziffer für eine EAN Nummer berechnen 

 

  

 
1 EAN steht für Europäische Artikelnummer (engl. European Acticle Number) 

4 VBA2-Quellcode der Anwendung 
 

Sub MainProc() 
    Const conTitle As String = "Prüfziffer f. EAN Nummer berechnen." 
    Dim intLen        As Integer       ' Länge einer EAN Nummer ohne Prüfziffer 
    Dim strEAN       As String         ' 12-stellige EAN-Nummer ohne Prüfziffer 
    Dim strReturn  As String * 1   ' Rückgabewert d. Funktionsprozedur GetCheckDigit 
    strEAN = "401234598765" 
    ' Gliederungszeichen aus strEAN entfernen. 
    strEAN = Replace(strEAN, "-", "") 
    strEAN = Replace(strEAN, " ", "") 
    intLen = Len(strEAN)            ' Länge der EAN Nummer ohne Prüfziffer 
    If intLen = 12 Then 
        strReturn = GetCheckDigit (strEAN) 
        If strReturn = vbNullString Then 
            MsgBox "Fehler bei d. Prüfziffernberechnung f. EAN-Nummer" 
        Else 
            MsgBox "Die Prüfziffer der EAN-Nummer = " & strEAN & " lautet " & strReturn & ".", vbInformation, conTitle 
        End If 
    Else 
        MsgBox "Länge der EAN-Nummer ist falsch!", vbCritical, conTitle 
    End If 
End Sub 

Listing 1: Subprozedur 'MainProc' 

Function GetCheckDigit (strEAN As String) As String 
    ' Aufgabe: Prüfziffer für eine 12-stellige EAN Nummer berechnen. 
    Dim intLoop      As Integer ' Schleifenzähler 
    Dim intSum       As Integer ' Produktsumme 
    Dim intLen        As Integer ' Länge d. übergebenen EAN-Nummer ohne Prüfziffer 
    Dim intWeight  As Integer ' Gewichtung einer Datenstelle der übergebenen EAN mit 1 oder 3 
    intLen = Len(strEAN) 
    intWeight = 1 
    ' Länge der EAN-Nummer ohne Prüfziffer 
    If intLen = 12 Then  
        For intLoop = 1 To intLen 
            intSum = intSum + CInt(Val(Mid$(strEAN, intLoop, 1))) * intWeight 
            intWeight = IIf(intWeight = 1, 3, 1) 
        Next intLoop 
        GetCheckDigit = 10 - intSum Mod 10 ' Prüfziffer berechnen 
    End If 
End Function 

Listing 2: Funktionsprozedur 'GetCheckDigit' 

 
2 VBA steht für Visual Basic for Applications, einer proprietären Skriptsprache für die Office-Suite der Firma 
Microsoft. 


